Les polynômes…le début

© Pierre Lantagne
Enseignant retraité du Collège de Maisonneuve

Ce document est une révision de celui produit en 2003. L'objectif principal de *Les polynômes…le début* est de transposer en Maple l'études des polynômes à coefficients réels sur le corps des nombres réels. Après une révision des termes propres au langage des polynômes, ce document présente le développement d'une polynôme en schéma complet de Horner. L'algorithme de la division synthétique est ensuite naturellement intégré à la procédure de décomposition en schéma de Horner. Ce document Maple se termine par l'énumération de quelques macro-commandes les plus courantes pour la manipulation d'expressions polynomiales.

Bonne lecture à tous !

* Ce document Maple est exécutable avec la version 2020.1

Initialisation

```
> restart;
> with(plots,display,pointplot,setoptions):
  setoptions(size=[400,400],axesfont=[times,roman,8]):
```

Définition d'un polynôme à coefficients réels

Soit $a_n, a_{n-1}, a_{n-2}, \ldots, a_2, a_1, a_0 \in \mathbb{R}$ et soit n est un entier non négatif, alors toute expression pouvant s'exprimer sous la forme

$$a_n x^n + a_{n-1} x^{n-1} + a_{n-2} x^{n-2} + \ \ldots \ + a_2 x^2 + a_1 x^1 + a_0 x^0$$

est appelée polynôme à une indéterminée à coefficients réels. On dit aussi polynôme à une seule variable.

Dans les développements polynomiales, on convient d'écrire x au lieu de x^1 et on convient de poser $x^0 = 1$, quelque soit $x \in \mathbb{R}$.

Lorsque l'indéterminée est x, nous dirons du polynôme que c'est un polynôme en x.

L'ensemble de tous les polynômes en x où $a_k \in \mathbb{R}$ est noté par $\mathbb{R}[x]$.

Par exemple, l'expression $-5x^4 + x^3 - 4x - 3 \in \mathbb{R}[x]$ car cette expression peut être réécrite sous la forme:

$$-5x^4 + 1x^3 + 0x^2 + (-4)x + (-3)$$

où $a_4 = -5$, $a_3 = 1$, $a_2 = 0$, $a_1 = -4$ et $a_0 = -3$.

Terminologie:

– les nombres a_k , $k = 0..n$ sont appelés *coefficients* du polynôme;

– le coefficient a_k de la plus haute puissance de x est appelé le *coefficient dominant* du polynôme;

– chaque produit $a_k x^k$ de la somme $a_n x^n + a_{n-1} x^{n-1} + \ldots + a_2 x^2 + a_1 x + a_0$ est un *terme* du polynôme;

– le terme a_0 est appelée *terme constant* du polynôme;

– si un coefficient a_k est nul, nous omettons d'écrire le terme $a_k x^k$;

– si tous les coefficients a_k, $k = 1..n$, sont tous nuls, de tels polynômes sont des *polynômes constants*;

– si tous les coefficients a_k, $k = 0..n$, sont tous nuls , le polynôme est appelé *polynôme nul*, et on le note *0*.

Polynôme	Coefficient dominant	Terme constant
$5x^7 + 4x^3 - 5x^2 + 6x + 16$	5	16
$-x^4 + 2x^3 - 4x^2 + 5x - 6$	-1	-6
$7x^5 - 2x^3 - x^2 + 6x$	7	0
$2x^7 + 3x^6 + x^8 - 2x - 5 + 2x^2$	1	-5

Puisque $\mathbb{R}[x]$ est l'ensemble des polynômes en x à coefficients réels, $\mathbb{Z}[x]$ désignera l'ensemble des polynômes en x à coefficients entiers et $\mathbb{Q}[x]$ l'ensemble des polynômes en x à coefficients rationnels. On a évidemment $\mathbb{Z}[x] \subseteq \mathbb{Q}[x] \subseteq \mathbb{R}[x]$.

• Pour Maple, un polynôme est une expression (un objet) qui a été créée en utilisant les opérateurs arithmétiques +, -, *, et ^, opérant des variables informatiques, c'est-à-dire des noms. Par exemple, l'assignation suivante crée un objet P de type polynôme.

```
> P:=2*x^7+3*x^6+x^8-2*x-5+2*x^2;
```

$$P := x^8 + 2x^7 + 3x^6 + 2x^2 - 2x - 5 \tag{2.1}$$

```
> type(P,polynom);
  Coefficient_dominant=lcoeff(P);
```

$$\mathit{true}$$

$$\mathit{Coefficient\_dominant} = 1 \tag{2.2}$$

Qu'en est-il de l'expression $(2x^3 - 3)^5$?

```
> P:=(2*x^3-3)^5;
```

$$P := (2x^3 - 3)^5 \tag{2.3}$$

```
> type(P,polynom);
  Coefficient_dominant=lcoeff(P);
```

$$\mathit{true}$$

$$\mathit{Coefficient\_dominant} = 32 \tag{2.4}$$

La macro-commande expand permet de réécrire $(2x^3 - 3)^5$ sous la forme développée $a_n x^n + a_{n-1} x^{n-1} + a_{n-2} x^{n-2} + \dots + a_2 x^2 + a_1 x + a_0$.

```
> P=expand(P);
```

$$(2x^3 - 3)^5 = 32x^{15} - 240x^{12} + 720x^9 - 1080x^6 + 810x^3 - 243 \tag{2.5}$$

L'expression $x^3 + 5x^2 + 11x + x^{-1}$ n'est pas un polynôme

```
> P:=x^3+5*x^2+11*x+x^(-1);
```

$$P := x^3 + 5x^2 + 11x + \frac{1}{x} \tag{2.6}$$

```
> type(P,polynom);
```

$$\mathit{false} \tag{2.7}$$

C'est aussi le cas pour l'expression $\sqrt{(2x-3)^2}$.

```
> P:=sqrt((2*x-3)^2);
```

$$P := \sqrt{(2x-3)^2} \tag{2.8}$$

```
> type(P,polynom);
```

$$\mathit{false} \tag{2.9}$$

En effet, P ne peut pas être exprimé sous la forme $a_n x^n + a_{n-1} x^{n-1} + a_{n-2} x^{n-2} + \ldots + a_2 x^2 + a_1 x + a_0$.

```
> P=simplify(P,assume=real);
```

$$\sqrt{(2x-3)^2} = \mathrm{signum}\left(x - \frac{3}{2}\right)(2x-3) \tag{2.10}$$

```
> ``=rhs(convert((2.10),piecewise));
```

$$= \begin{cases} -2x+3 & x \le \frac{3}{2} \\ 2x-3 & \frac{3}{2} < x \end{cases} \tag{2.11}$$

Ce qui montre que l'on a $\sqrt{(2x-3)^2} = |2x-3|$.

• Dans ce cours, nous nous intéresserons qu'aux polynômes à une seule indéterminée, toutes autres variables dans un polynôme sont traitées comme des constantes. Par exemple:

```
> P:=x^5-a^5;
```

$$P := -a^5 + x^5 \tag{2.12}$$

```
> type(P,polynom);
```

$$\mathit{true} \tag{2.13}$$

En fait, Maple traite ici P comme un polynôme de deux variables à moins de lui signifier que a est du type constant. Dans ce cas, P sera reconnu par Maple comme un polynôme d'une seule indéterminée.

Obtenons la liste des indéterminées du polynôme P.

```
> indets(P);
```

$$\{a, x\} \tag{2.14}$$

Afin que Maple reconnaisse P comme un polynôme d'une seule indéteminée, sauvegardons d'abord la liste système constants des constantes Maple dans Liste afin de la restituer par la suite.

```
> Liste:=constants;
```

$$\mathit{Liste} := \mathit{false}, \gamma, \infty, \mathit{true}, \mathit{Catalan}, \mathit{FAIL}, \pi \tag{2.15}$$

Ajoutons ensuite « a » à la liste sytème pour que « a » soit ensuite traité comme telle par Maple.

```
> constants:=constants,a;
```

$$\mathit{constants} := \mathit{false}, \gamma, \infty, \mathit{true}, \mathit{Catalan}, \mathit{FAIL}, \pi, a \tag{2.16}$$

Obtenons maintenant l'ensemble des indéterminées du polynôme P.

```
> indets(P);
```

$$\{x\} \tag{2.17}$$

Voilà, P est toujours reconnu par Maple comme un polynôme mais nous savons maintenant qu'il est reconnu comme un polynôme d'une seule indéterminée.

```
> type(P,polynom);
```

$$true \tag{2.18}$$

Pour la suite de ce document, revenons à la liste sytème des constantes Maple. P sera à nouveau reconnu comme un polynôme à deux indéterminées.

```
> constants:=Liste;
```

$$constants := false, \gamma, \infty, true, Catalan, FAIL, \pi \tag{2.19}$$

```
> indets(P);
```

$$\{a, x\} \tag{2.20}$$

Degré d'un polynôme

Soit un polynôme P non nul. Si n est le plus grand entier tel que $a_n \neq 0$, alors on dira du polynôme P = $a_n x^n + a_{n-1} x^{n-1} + a_{n-2} x^{n-2} + \ldots + a_2 x^2 + a_1 x + a_0$
qu'il est de degré n. Ce qui est noté par d°(P) $= n$

```
> P := 5*x^3-x^2+3*x-7;
```

$$P := 5x^3 - x^2 + 3x - 7 \tag{3.1}$$

```
> `d°`('P')=degree(P,x);
  Coeff_dominant=lcoeff(P);
```

$$d°(P) = 3$$

$$Coeff\_dominant = 5 \tag{3.2}$$

Le degré d'un polynôme constant c (autre que le polynôme nul) est 0 puisque l'on considère que le terme constant est le coefficient de x^0. Rappelons que l'on a convenu que $x^0 = 1$ et de ne pas l'écrire. Ainsi, nous avons $c = cx^0$, $c \neq 0$.

```
> P := -sqrt(3);
```

$$P := -\sqrt{3} \tag{3.3}$$

```
> `d°`('P')=degree(P,x);
  Coefficient_dominant=lcoeff(P);
```

$$d°(P) = 0$$

$$Coefficient\_dominant = -\sqrt{3} \tag{3.4}$$

Par la définition même du degré d'un polynôme, le degré du polynôme nul n'est donc pas définie. Tout de même, on convient que le degré du polynôme nul est $-\infty$. Cette convention trouvera plus loin sa justification dans l'arithmétique des polynômes.

```
> P := 0;
```

$$P := 0 \tag{3.5}$$

```
> degree(P,x);
```

$$-\infty \tag{3.6}$$

Arithmétique des polynômes

Deux polynômes sont égaux si, par définition, ils ont le même degré et que tous les coefficients correspondants sont égaux.

On peut additionner, soustraire, multiplier et diviser deux polynômes en appliquant les propriétés dans $\mathbb{R}$ de l'addition et de la multiplication.

L'addition de deux polynômes $x^3 + 2x^2 - 5x + 7$ et $x^3 - 5x^2 + 3$ est effectuée en additionnant les coefficients correspondants. Obtenons leur somme:

$$
\begin{aligned}
(x^3 + 2x^2 - 5x + 7) + (x^3 - 5x^2 + 3) &= x^3 + 2x^2 - 5x + 7 + x^3 - 5x^2 + 3 \\
&= (1 + 1)\, x^3 + (2 - 5)\, x^2 + (-5 + 0)\, x + (7 + 3) \\
&= 2x^3 - 3x^2 - 5x + 10
\end{aligned}
$$

Obtenons maintenant la différence $(x^3 + 2x^2 - 5x + 7) - (x^3 - 5x^2 + 3)$.

$$
\begin{aligned}
(x^3 + 2x^2 - 5x + 7) - (x^3 - 5x^2 + 3) &= x^3 + 2x^2 - 5x + 7 - x^3 + 5x^2 - 3 \\
&= (1 - 1)\, x^3 + (2 + 5)\, x^2 + (-5 + 0)\, x + (7 - 3) \\
&= 7x^2 - 5x + 4
\end{aligned}
$$

En convenant que le degré du polynôme nul est $-\infty$, on peut énoncer à l'égard de l'addition et de la soustraction de polynômes que le degré du polynôme somme est toujours inférieure ou égale au maximum des degrés de chaque polynôme. Pour P1, P2 $\in \mathbb{R}[x]$, nous avons que

$$d°(P1 \pm P2) \leq \max(\, d°(P1),\, d°(P2)\,)$$

Transposons en Maple cette inégalité et vérifions-là avec quelques exemples. Cette vérification sera faite en évaluant la valeur de vérité de l'inégalité avec la macro-commande `evalb`.

```
> P1 := x^3+2*x^2-5*x+7;
  P2 := x^3-5*x^2+3;
```

$$P1 := x^3 + 2x^2 - 5x + 7$$

$$P2 := x^3 - 5x^2 + 3 \qquad \textbf{(4.1)}$$

```
> 'P1'+'P2' = P1+P2;
  'P1'-'P2' = P1-P2;
```

$$P1 + P2 = 2x^3 - 3x^2 - 5x + 10$$

$$P1 - P2 = 7x^2 - 5x + 4 \qquad \textbf{(4.2)}$$

```
> evalb(degree(P1+P2)<=max(degree(P1),degree(P2)));
  evalb(degree(P1-P2)<=max(degree(P1),degree(P2)));
```

true

true **(4.3)**

```
> P1 := 4*x^5-2*x^2-x+5;
  P2 := 6*x^5+2*x-x+5;
```

$$P1 := 4x^5 - 2x^2 - x + 5$$

$$P2 := 6x^5 + x + 5 \qquad \textbf{(4.4)}$$

```
> 'P1'+'P2' = P1+P2;
```

```
'P1'-'P2' = P1-P2;
```
$$P1 + P2 = 10\,x^5 - 2\,x^2 + 10$$
$$P1 - P2 = -2\,x^5 - 2\,x^2 - 2\,x \quad \textbf{(4.5)}$$

```
> evalb(degree(P1+P2)<=max(degree(P1),degree(P2)));
  evalb(degree(P1-P2)<=max(degree(P1),degree(P2)));
```
$$true$$
$$true \quad \textbf{(4.6)}$$

```
> P1 := x+1;
  P2 := x+1;
```
$$P1 := x + 1$$
$$P2 := x + 1 \quad \textbf{(4.7)}$$

```
> 'P1-P2' = P1-P2;
```
$$P1 - P2 = 0 \quad \textbf{(4.8)}$$

```
> evalb(degree(P1-P2)<=max(degree(P1),degree(P2)));
```
$$true \quad \textbf{(4.9)}$$

La propriété de la distributivité (à gauche et à droite) de la multiplication sur l'addition dans $\mathbb{R}$ est appliquée pour développer la multiplication de deux polynômes. Multiplions les polynômes $x^3 + 2\,x^2 - 5\,x + 7$ et $x^3 - 5\,x^2 + 3$.

$$\left(x^3 + 2\,x^2 - 5\,x + 7\right) \times \left(x^3 - 5\,x^2 + 3\right) = \left(x^3 + 2\,x^2 - 5\,x + 7\right) \times \left(x^3\right) - \left(x^3 + 2\,x^2 - 5\,x + 7\right) \times \left(5\,x^2\right) + \left(x^3 + 2\,x^2 - 5\,x + 7\right) \times (3)$$
$$= x^6 + 2\,x^5 - 5\,x^4 + 7\,x^3 - 5\,x^5 - 10\,x^4 + 25\,x^3 - 35\,x^2 + 3\,x^3 + 6\,x^2 - 15\,x + 21$$
$$= x^6 - 3\,x^5 - 15\,x^4 + 35\,x^3 - 29\,x^2 - 15\,x + 21$$

```
> P1:=x^3+2*x^2-5*x+7:
  P2:=x^3-5*x^2+3:
  P1*P2=expand(P1*P2);
```
$$\left(x^3 + 2\,x^2 - 5\,x + 7\right)\left(x^3 - 5\,x^2 + 3\right) = x^6 - 3\,x^5 - 15\,x^4 + 35\,x^3 - 29\,x^2 - 15\,x + 21 \quad \textbf{(4.10)}$$

Le degré du produit est égal à la somme des degrés si aucun des polynômes n'est le polynôme nul.

$$\mathrm{d°(P1 \cdot P2) = d°(P1) + d°(P2)}, \quad \text{si P1} \neq 0 \text{ et si P2} \neq 0$$

```
> P1 := 6*x^5-2*x^2-x+5;
  P2 := 6*x^4+2*x-x+5;
```
$$P1 := 6\,x^5 - 2\,x^2 - x + 5$$
$$P2 := 6\,x^4 + x + 5 \quad \textbf{(4.11)}$$

```
> evalb(degree(P1*P2)=degree(P1)+degree(P2));
```
$$true \quad \textbf{(4.12)}$$

```
> P1 := 6*x^5-2*x^2-x+5;
  P2 := -5;
```
$$P1 := 6\,x^5 - 2\,x^2 - x + 5$$
$$\textbf{(4.13)}$$

$$P2 := -5 \tag{4.13}$$

```
> evalb(degree(P1*P2)=degree(P1)+degree(P2));
```

$$true \tag{4.14}$$

La division de deux polynômes mérite une attention paticulière. La division à considérer est la **division euclidienne** et non pas la division décimale. La division euclidienne est la première division que l'on apprend à l'école primaire. Par exemple, soit la division $1339 \div 17$. On exprime alors, manu scriptus, cette division comme suit

$$\begin{array}{r|l}
\text{dividende} \longrightarrow 1339 & 17 \longleftarrow \text{diviseur} \\
\underline{119} & 78 \longleftarrow \text{quotient} \\
149 & \\
\underline{136} & \\
13 \longleftarrow \text{reste} &
\end{array}$$

La division $1339 \div 17$ donne le quotient 78 et le reste 13

$$1339 = 17 \text{ x } 78 + 13$$

On identifie le nombre 1339 comme étant le **dividende**, 17 le **diviseur**, 78 le **quotient** et 13 le **reste**.

(**dividende** = **diviseur** x **quotient** + **reste**)

La division euclidienne met en évidence quatre entités: *dividende*, *diviseur*, *quotient* et *reste*. Les termes dividende, diviseur, quotient et reste sont également employés dans la division polynomiale.

Pour effectuer la division euclidienne de deux entiers, nous devons avoir le dividende supérieur ou égal au diviseur. La division euclidienne est terminée lorsque le reste est inférieur au diviseur. Similairement pour effectuer la division polynomiale, nous devons avoir le degré du polynôme dividende supérieur ou égal à celui du polynôme diviseur et la division polynomiale est terminée lorsque le degré du polynôme reste est inférieur à celui du polynôme diviseur.

d°(Polynôme dividende) $\geq$ d°(Polynôme diviseur) où Polynôme diviseur $\neq$ 0 (Polynôme nul)

Nous devons avoir le degré du polynôme reste inférieur au degré du polynôme diviseur (même si le polynôme reste est le polynôme nul car son degré est $-\infty$).

d°(Polynôme reste) < d°(Polynôme diviseur)

Divisons le polynôme $x^4 - 3x^3 + 12x - 7$ par le polynôme $x^2 + x - 1$ que nous notons $(x^4 - 3x^3 + 12x - 7) \div (x^2 + x - 1)$.

$$\begin{array}{l|l}
x^4 - 3x^3 \qquad + 12x - 7 & x^2 + x - 1 \\
\underline{x^4 + x^3 - x^2} & x^2 - 4x + 5 \\
\quad -4x^3 + x^2 + 12x - 7 & \\
\quad \underline{-4x^3 - 4x^2 + 4x} & \\
\qquad\qquad 5x^2 + 8x - 7 & \\
\qquad\qquad \underline{5x^2 + 5x - 5} & \\
\qquad\qquad\qquad 3x - 2 &
\end{array}$$

$$(x^4 - 3x^3 + 12x - 7) = (x^2 + x - 1) \cdot (x^2 - 4x + 5) + (3x - 2)$$

(dividende = diviseur x quotient + reste)

Soit P_1 et P_2 deux polynômes de degrés non nuls où d°$(P_1) \geq$ d°(P_2). Alors, il existe deux polynômes uniques Q et R tels que

$$P_1 = P_2\,Q + R$$

où d°(R) < d°(P_2).

Q est appelé le polynôme quotient tandis que **R** est appelé le **polynôme reste**.

Les polynômes $\boldsymbol{P_1}$ et $\boldsymbol{P_2}$ sont appelés respectivement le **polynôme dividende** et **polynôme diviseur**.

Soit les polynômes $P_1 = x^4 - 16$ et $P_2 = x^2 + 3x + 1$. Obtenons d'abord le quotient et le reste de la division $P_1 \div P_2$. Utilisons les macro-commandes de la bibliothèque de base quo (quotient) et rem (reste).

```
> P1:=x^4-16;
  P2:=x^2+3*x+1;
```

$$P1 := x^4 - 16$$

$$P2 := x^2 + 3x + 1 \tag{4.15}$$

Obtenons le quotient de la division $P_1 \div P_2$.

```
> quo(P1,P2,x);
```

$$x^2 - 3x + 8 \tag{4.16}$$

Obtenons le reste de la division $P_1 \div P_2$.

```
> rem(P1,P2,x);
```

$$-21x - 24 \tag{4.17}$$

Exprimons finalement la division $P_1 \div P_2$ sous la forme algébrique de la division.

```
> P1/P2=quo(P1,P2,x)+rem(P1,P2,x)/P2;
```

$$\frac{x^4 - 16}{x^2 + 3x + 1} = x^2 - 3x + 8 + \frac{-21x - 24}{x^2 + 3x + 1} \tag{4.18}$$

Exprimons la division $P_1 \div P_2$ sous la forme de l'algorithme de la division.

```
> P1=P2*quo(P1,P2,x)+(rem(P1,P2,x));
```

$$x^4 - 16 = (x^2 + 3x + 1)\,(x^2 - 3x + 8) - 21x - 24 \tag{4.19}$$

On a donc un polynôme reste non nul $R = -21x - 24$ dont le degré est inférieur à celui du polynôme diviseur. En effet

```
> evalb(degree(rem(P1,P2,x))<degree(P2));
```

$$\mathit{true} \tag{4.20}$$

On peut également exprimer la division $P_1 \div P_2$ sous la forme de l'algorithme de la division sans utiliser explicitement la macro-commande `rem`. Dans ce cas, il faut préciser, en option, dans la macro-commande `quo`, une variable **libre** qui est automatiquement affectée du reste de la division.

```
> P1=P2*quo(P1,P2,x,'Reste')+Reste;
```

$$x^4 - 16 = (x^2 + 3x + 1)\,(x^2 - 3x + 8) - 21x - 24 \tag{4.21}$$

Lorsque le reste est nul, comme par exemple pour la division 27 ÷ 3, nous disons que 3 divise 27 et on écrit « 3 | 27 ». Idem dans le cas de la division polynomiale. Si on affirme qu'un polynôme P2 *de degré non nul* divise un polynôme P1, c'est que le reste de la division P1 ÷ P2 est nul.

L'algorithme euclidien de la division peut donc être utilisé pour déterminer si un polynôme donné P2 est un facteur d'un autre polynôme P1.

Par exemple, déterminons si P2 = $2x^2+1$ est un facteur de P1 = $6x^3-4x^2+3x-2$. Obtenons directement le reste de P1 $\div$ P2 avec la macro-commande rem.

```
> P1:=6*x^3-4*x^2+3*x-2;
  P2:=2*x^2+1;
```

$$P1 := 6x^3-4x^2+3x-2$$

$$P2 := 2x^2+1 \tag{4.22}$$

```
> rem(P1,P2,x);
```

$$0 \tag{4.23}$$

Le reste nul montre donc que $(2x^2+1)$ est un facteur du polynôme $(6x^3-4x^2+3x-2)$.

L'égalité suivante montre bien que P2 | P1.

```
> P1/P2=quo(P1,P2,x,'r')+r;
```

$$\frac{6x^3-4x^2+3x-2}{2x^2+1} = 3x-2 \tag{4.24}$$

Évaluation d'un polynôme

La macro-commande eval de la bibliothèque principale permet une évaluation rapide d'un polynôme avec une valeur partriculière de l'indéterminée. Par exemple, évaluons le polynôme

$\sqrt{2}\,x^7-3\sqrt{2}\,x^5-\frac{2x^6}{3}-5x^4+19x^2-6x^3+18x+6$ avec $x=2$.

```
> P:=sqrt(2)*x^7-3*sqrt(2)*x^5-2/3*x^6-5*x^4+19*x^2-6*x^3+18*x+6;
```

$$P := \sqrt{2}\,x^7-3\sqrt{2}\,x^5-\frac{2x^6}{3}-5x^4+19x^2-6x^3+18x+6 \tag{5.1}$$

```
> eval(P,x=2);
```

$$32\sqrt{2}-\frac{158}{3} \tag{5.2}$$

À l'aide de la forme inerte de la macro-commande `eval`, documentons la zone des résultats en posant l'égalité suivante.

```
> Eval('P',x=2)=eval(P,x=2);
```

$$P\Big|_{x=2} = 32\sqrt{2}-\frac{158}{3} \tag{5.3}$$

Il est d'usage en mathématique d'employer la **notation** P(c) pour exprimer l'évaluation d'un polynôme P avec x = c.

```
> 'P'(2)=eval(P,x=2);
```

$$P(2) = 32\sqrt{2}-\frac{158}{3} \tag{5.4}$$

ATTENTION: La notation P(2) est un raccourci et indique seulement qu'il s'agit d'évaluer le polynôme P avec $x=2$. Il ne faut pas considérer P(2) comme étant l'image de 2 par la fonction P car P est un polynôme, P n'est pas une fonction. Par contre, on peut très bien définir une fonction f dont la règle f(x) est un certain polynôme P. On dit alors que f est une fonction polynomiale.

Dans le cas où P(c) = 0, on dit du nombre « c » qu'il est une racine du polynôme P. Montrons que $\sqrt{3}$ est une racine du polynôme P.

```
> 'P'(sqrt(3))=eval(P,x=sqrt(3));
```

$$P(\sqrt{3}) = 0 \tag{5.5}$$

REMARQUE: Dans le language des fonctions, si $f(c) = 0$, on dit effectivement que c est un zéro de la fonction f. Il est important de faire cette distinction langagière.

Algorithme de la division (division euclidienne)

Un cas intéressant que nous allons approfondir est celui dont le diviseur est un polynôme du premier degré de la forme $(x - a)$. Dans ce cas, l'algorithme de la division établit que, si $(x - a)$ est un diviseur d'un polynôme P, alors le reste R est 0 et donc que $(x - a)$ est un facteur. Si $(x - a)$ ne divise pas P, le reste n'est donc pas nul. Puisque que le degré du reste est plus petit que le degré de $(x - a)$, le reste est donc de degré 0 ou $-\infty$. Cela signifie que, dans ce cas, le reste est donc un nombre non nul ou 0.

En résumé, la division $P \div (x - a)$ exprimée sous la forme de l'algorithme de la division est:

$$P = (x - a)\,Q + r$$

où le reste r est un nombre réel pouvant être nul. Ainsi, en substituant $x = a$ dans l'algorithme de la division, nous obtenons

$$\begin{aligned} P(a) &= (a - a)\,Q(a) + r \\ &= 0\,Q(a) + r \\ &= 0 + r \\ &= r \end{aligned}$$

Ce qui montre que si un polynôme P est divisé par $(x - a)$, alors le reste r = P(a). **(Théorème du reste**).

Ce qui est intéressant à propos du théorème du reste c'est la proposition suivante:

$(x - a)$ est un facteur d'un polynôme P si et seulement si P$(a) = 0$.

Posons-nous la question suivante: $(x + 2)$ est-il un facteur du polynôme P = $6x^5 - 20x^3 - 7x + 10$? Pour répondre à cette question, il suffit d'obtenir le reste de la division de $6x^5 - 20x^3 - 7x + 10$ par $(x + 2)$.

```
> P:=12*x^5-4*x^4-33*x^3+38*x^2-15*x+2;
```

$$P := 12x^5 - 4x^4 - 33x^3 + 38x^2 - 15x + 2 \tag{6.1}$$

```
> 'Le_reste'='P'(-2);
  ``=eval(P,x=-2);
```

$$\begin{aligned} Le\_reste &= P(-2) \\ &= 0 \end{aligned} \tag{6.2}$$

Puisque P$(-2) = 0$, on peut affirmer que $(x + 2)$ divise le polynôme P et est donc un facteur de P. Le théorème du reste est donc fort utile pour déterminer assez rapidement si $(x - a)$ est un facteur d'un polynôme.

La prochaine section décrit ce qu'est le schéma de Horner d'un polynôme. Dans le cas où le polynôme diviseur est du premier degré de la forme $(x - a)$, l'écriture d'un polynôme en schéma de Horner est très utile pour diviser plus rapidement. La division réalisé à l'aide du schéma de Horner est appelée division synthétique.

Schéma de Horner d'un polynôme

Le développement d'un polynôme en **schéma de Horner** évite l'utilisation des puissances pour l'indéterminée. La réécriture du polynôme peut se limiter alors aux seules opérations de multiplication et d'addition.

Par exemple, le développement en schéma de Horner de $-2x^3 + 4x^2 + x + 12$ est

$$((-2x + 4)x + 1)x + 12$$

D'une façon générale, un polynôme P, tel que

$$P = a_n x^n + a_{n-1} x^{n-1} + a_{n-2} x^{n-2} + \cdots + a_2 x^2 + a_1 x + a_0$$

peut être construit au moyen de la suite de polynômes de récurrence

$$P_{k+1} = x P_k + a_{n-k-1} \text{ , pour } k = 0 \ldots n - 1 \text{ avec } P_0 = a_n$$

Ce qui donne, par récurrence, le développement :

$$P = \left(a_n x^{n-1} + a_{n-1} x^{n-2} + a_{n-2} x^{n-3} + \cdots + a_2 x + a_1\right) x + a_0$$
$$P = \left(\left(a_n x^{n-2} + a_{n-1} x^{n-3} + a_{n-2} x^{n-4} + \cdots + a_2\right) x + a_1\right) x + a_0$$
$$\vdots$$
$$P = ((((\ldots(a_n)\, x + a_{n-1})\, x + a_{n-2})\, x + \ldots + a_2)\, x + a_1)\, x + a_0$$

La procédure suivante, basée sur la formule de récurrence $P_{k+1} = x P_k + a_{n-k-1}$, illustre la construction du schéma de Horner pour un polynôme de degré n quelconque.

```
> horner:=proc(n::nonnegint)
   local k,r,P;
   r:=n;
   P[0]:=a[r];
    if r=0 then a[r]
    else
     for k from 0 by 1 while k<r do
      P[k+1]:=x*P[k]+a[r-k-1]
     od
    fi
  end proc:
```

```
> horner(0);
  horner(1);
  horner(2);
```

$$a_0$$
$$x a_1 + a_0$$
$$x\left(x a_2 + a_1\right) + a_0 \qquad \textbf{(7.1)}$$

À l'aide de la procédure "horner", obtenons les développements de Horner pour des polynômes quelconques de degré inférieur ou égal à 8

```
> seq(print(horner(n)),n=0..8);
```

$$a_0$$
$$x a_1 + a_0$$
$$x\left(x a_2 + a_1\right) + a_0$$

$$x\left(x\left(x a_3+a_2\right)+a_1\right)+a_0$$
$$x\left(x\left(x\left(x a_4+a_3\right)+a_2\right)+a_1\right)+a_0$$
$$x\left(x\left(x\left(x\left(x a_5+a_4\right)+a_3\right)+a_2\right)+a_1\right)+a_0$$
$$x\left(x\left(x\left(x\left(x\left(x a_6+a_5\right)+a_4\right)+a_3\right)+a_2\right)+a_1\right)+a_0$$
$$x\left(x\left(x\left(x\left(x\left(x\left(x a_7+a_6\right)+a_5\right)+a_4\right)+a_3\right)+a_2\right)+a_1\right)+a_0$$
$$x\left(x\left(x\left(x\left(x\left(x\left(x\left(x a_8+a_7\right)+a_6\right)+a_5\right)+a_4\right)+a_3\right)+a_2\right)+a_1\right)+a_0 \quad \textbf{(7.2)}$$

Modifions légèrement cette procédure. Faisons passer, en paramètre dans la procédure maison horner, le polynôme à schématiser. Aussi, **reformulons à rebours** le schéma précédent. Dans la prochaine section, cela va nous permettre de mieux exposer la division synthétique de Horner.

```
> horner:=proc(p::polynom)
   local A,a,i,k,n,P;
   if degree(p)=0 or degree(p)=-infinity then p else
   A:=[seq(coeff(p,x,k),k=0..degree(p))];
   n:=nops(A);
    P[1]:=A[n];
    for k from 1 by 1 to n-1 do
    P[k+1]:=P[k]*x+A[n-k] od;
   sort(%,order=tdeg(x))
  fi
  end proc:
```

Obtenons, avec cette version de la procédure de horner, le schéma de Horner d'un polynôme quelconque de degré 7.

```
> P:=sum(a[k]*x^k,k=0..7);
```

$$P := a_7 x^7+a_6 x^6+a_5 x^5+a_4 x^4+a_3 x^3+a_2 x^2+x a_1+a_0 \quad \textbf{(7.3)}$$

```
> 'P'=horner(P);
```

$$P=\left(\left(\left(\left(\left(\left(a_7 x+a_6\right) x+a_5\right) x+a_4\right) x+a_3\right) x+a_2\right) x+a_1\right) x+a_0 \quad \textbf{(7.4)}$$

```
> ``=simplify(rhs((7.4)));
```

$$=a_7 x^7+a_6 x^6+a_5 x^5+a_4 x^4+a_3 x^3+a_2 x^2+x a_1+a_0 \quad \textbf{(7.5)}$$

L'argument de la procédure "horner" ne requiert pas que le polynôme soit donné en ordre décroissant d'exposants.

Remarque: Lorsque certains coefficients sont nuls dans le polynôme à schématiser, il est nécessaire de contourner le mécanisme de la simplification automatique afin d'avoir un schéma complet de Horner. Modifions la procédure "horner" par l'ajout d'une instruction dans la procédure qui substitura chaque coefficient nul du polynôme par une même variable libre. Choisissons la variable « O » majuscule pour simuler un coefficient nul dans le shéma complet de Horner.

Version finale de la procédure horner.

```
> horner:=proc(p::polynom)
   global O;
   local A,k,n,P;
   if degree(p)=0 or degree(p)=-infinity then p else
   A:=[seq(coeff(p,x,k),k=0..degree(p))];
   n:=nops(A);
```

```
    P[1]:=A[n];
    for k from 1 to n-1 do
     if A[n-k]=0 then A[n-k]:=O fi;
     P[k+1]:=P[k]*x+A[n-k];
    od;
  sort(%,order=tdeg(x));
  fi;
  end proc:
```

Soit le polynône $x^6 - 2x^4 + 4x^2 + 11$.

```
> P:=x^6-2*x^4+4*x^2+11;
```

$$P := x^6 - 2x^4 + 4x^2 + 11 \quad (7.6)$$

Voici alors le schéma complet de Horner du polynôme $x^6 - 2x^4 + 4x^2 + 11$.

```
> P=horner(P);
```

$$x^6 - 2x^4 + 4x^2 + 11 = (((((x+\mathrm{O})\,x-2)\,x+\mathrm{O})\,x+4)\,x+\mathrm{O})\,x+11 \quad (7.7)$$

Vérifions en simplifiant ce schéma sur la base de l'égalité $\mathrm{O} = 0$.

```
> ``=simplify(rhs((7.7)),[O=0]);
```

$$= x^6 - 2x^4 + 4x^2 + 11 \quad (7.8)$$

L'intérêt à développer un polynôme en schéma de Horner apparaîtra clairement dans la division polynomiale lorsque le diviseur est un polynôme du premier degré de la forme $(x - a)$.

Division synthétique ou méthode de Horner

Lorsque le diviseur est du premier degré de la forme $(x - a)$, voyons comment les polynômes de récurrence P_{k+1} sont utiles dans la division polynomiale pour obtenir les coefficients du **polynôme quotient** ainsi que le **reste**.

Soit P un polynôme d'indéterminée x et soit $(x - a)$ le diviseur. L'algorithme de la division permet d'écrire l'égalité

$$P = (x - a)\,Q + r$$

avec $r = 0$ ou si $r \neq 0$, $\mathrm{d}°(r) = 0$.

Notons symboliquement un polynôme $a_n x^n + a_{n-1} x^{n-1} + a_{n-2} x^{n-2} + ... + a_2 x^2 + a_1 x + a_0$ par un n-uplet composé de ses coefficients $(a_n, a_{n-1}, ..., a_2, a_1, a_0)$. La division euclidienne de P par $x - a$ donne un polynôme quotient Q de degré $n - 1$ qu'on peut écrire symboliquement comme suit

$$Q = \left(q_{n-1}, q_{n-2}, ..., q_2, q_1, q_0\right) \text{ où } q_{n-1} = a_n$$

Cette façon de noter un polynôme exclusivement sur la base de ses coefficients va nous permettre d'exprimer les calculs successifs effectués au cours de la division polynomiale.

À la première étape de la division, on a l'égalité $P = (x - a)\,Q_1 + R_1 = (1, a)\,q_{n-1} + R_1$ où $q_{n-1} = a_n$ et R_1 un premier polynôme reste dont le degré est inférieur ou égal à $n - 1$.

$$q_{n-1} = a_n$$

De même, à la deuxième étape, on a l'égalité $P = (x - a)\,Q_2 + R_2 = (1, a)\left(q_{n-1}, q_{n-2}\right) + R_2$ où $q_{n-2} = a\,a_n + a_{n-1}$ et R_2 est un deuxième polynôme dont le degré est inférieur ou égal à $n - 2$.

$$q_{n-2} = a\,q_{n-1} + a_{n-1}$$

À la troisième étape, on a l'égalité $P = (x - a)\,Q_3 + R_3 = (1, a)\left(q_{n-1}, q_{n-2}, q_{n-3}\right) + R_3$ où

$q_{n-3} = a\left(a\,a_n + a_{n-1}\right) + a_{n-2}$ et R_3 est un troisième polynôme dont le degré est inférieur ou égal à $n-3$. On a donc que

$$q_{n-3} = a\,q_{n-2} + a_{n-2}$$

D'une manière générale, la formule de récurrence donnant q_{n-k} du polynôme quotient $\left(0, q_{n-1}, q_{n-2}, ..., q_2, q_1, q_0\right)$ est la formule

$$q_{n-k} = a\,q_{n-k+1} + a_{n-k+1}, \text{ pour } k = 1, 2, ..., n-1, n$$

Or, il est facile de montrer que

$$q_{n-k} = P_{k-1}\Big|_{x=a}, \text{ pour } k = 1, 2, ..., n-1, n$$

La formule de récurrence calculant les coefficients du polynôme quotient correspond, pour chaque itération, à l'évaluation d'un polynôme de récurrence de Horner avec $x = a$. De plus, notons que si on pose $k = n+1$ dans la formule de récurrence q_{n-k}, la notation $q_{-1} = P_n(a)$ représente le reste de la division.

Construisons donc une petite procédure prenant en charge les polynômes de récurrence de Horner laquelle évaluera, à chaque itération, avec $x = a$.

```
> div_syn:=proc(p::polynom,var::name,x0::complexcons)
   local A,a,k,n,Q,x;
   x:=var;
   if degree(p)=-infinity then
    printf(" ");
    printf("Le reste de la division de %a par (%a) est %a",p,x-x0,p);
    printf("\nLe polynôme quotient est 0")
   else
    A:=[seq(coeff(p,x,k),k=0..degree(p))];
    n:=nops(A);
    Q[1]:=A[n];
    for k from 1 to n-1 do
     Q[k+1]:=eval(Q[k]*x0+a[n-k-1],a[n-k-1]=A[n-k])
     od;
   printf(" ");
   printf("\n  Le reste de la division de (%a) par (%a) est %a",p,x-x0,sort(Q
  [k]));
   printf("\n  Le polynôme quotient est (%a)",sort(eval(sum(Q['k']*x^'n-k-1',
  'k'=1..n-1))))
   fi
  end proc:
```

Soit P = $6x^5 - 20x^3 - 7x + 10$. Divisons P par $(x+2)$ en appliquant la division synthétique (méthode de Horner).

```
> P:=6*x^5-20*x^3-7*x+10:
  div_syn(P,x,-2);

  Le reste de la division de (6*x^5-20*x^3-7*x+10) par (x+2) est -8
  Le polynôme quotient est (6*x^4-12*x^3+4*x^2-8*x+9)
```

Comme exemple d'application, déterminons toutes les valeurs de k afin que le polynôme $P = kx^3 + x^2 + k^2x + 3k^2 + 11$ soit divisible par $(x+2)$.

Obtenons le reste de la division de P par $(x+2)$ de deux manières: avec la division synthétique et avec le calcul de P(-2) avec la macro-commande `eval`.

```
> P:=k*x^3+x^2+k^2*x+3*k^2+11:
  div_syn(P,x,-2);
```

```
Le reste de la division de (k*x^3+k^2*x+3*k^2+x^2+11) par (x+2) est k^2-8*k+15
Le polynôme quotient est (k*x^2+k^2+4*k+(-2*k+1)*x-2)
```

```
> 'P'(-2)=eval(P,x=-2);
```

$$P(-2) = k^2 - 8k + 15 \quad \textbf{(8.1)}$$

Il suffit maintenant de trouver toutes les valeurs de *k,* si elles existent, qui annuleront le reste $-8k + 15 + k^2$. Ainsi, avec ces valeurs et par le théorème du reste, on affirmera que le polynôme P est divisible par $(x - 2)$.

```
> Sol:=solve(-8*k+15+k^2=0,{k});
```

$$Sol := \{k = 5\}, \{k = 3\} \quad \textbf{(8.2)}$$

Donc, avec $k = 5$ ou avec $k = 3$, le polynôme P est divisible par $x + 2$ puisqu'avec chacune de ces valeurs de *k*, le reste est nul.

Lorsqu'il s'agit d'obtenir le reste lors d'un calcul *manu scriptus*, la division synthétique est plus efficace qu'avec une évaluation directe de $P(a)$. La division synthétique reposant sur le schéma complet de Horner de P, voici ce qui montrera que l'évaluation un polynôme à partir de son schéma de Horner est en effet plus efficace que de l'évaluer par calcul direct: le nombre d'opérations impliquées est beaucoup moindre.

```
> with(codegen,cost);
```

$$[cost] \quad \textbf{(8.3)}$$

```
> P[A]:=sort(sum((k+1)*x^k,k=0..7));
  P[B]:=horner(P[A]);
```

$$P_A := 8x^7 + 7x^6 + 6x^5 + 5x^4 + 4x^3 + 3x^2 + 2x + 1$$

$$P_B := (((((((8x + 7)x + 6)x + 5)x + 4)x + 3)x + 2)x + 1 \quad \textbf{(8.4)}$$

```
> cost(P[A]);
  cost(P[B]);
```

$$7\,additions + 28\,multiplications$$

$$7\,additions + 7\,multiplications \quad \textbf{(8.5)}$$

Ce qui est intéressant avec la division synthétique (avec l'ordinateur ou avec un calcul *manu scriptus*), c'est que cette méthode révelle en même temps les coefficients du polynôme quotient ainsi que la valeur du reste. Comme nous le verrons, lorsque le reste sera non nul, la connaissance des coefficients du polynôme quotient peut être utile dans la recherche des autres racines d'un polyôme.

Manipulation d'expressions polynomiales

Voici une liste partielle de plusieurs macro-commandes Maple utiles pour manipuler des expressions polynomiales. On retrouvera une liste plus exhaustive en cliquant sur polynôme.

Macro-commandes utilitaires

- `coeff` pour extraire un coefficient dans un polynôm
- `coeffs` pour créer la séquence des coefficients d'un polynômes
- `degree` pour obtenir le degré d'un polynôme
- `lcoeff` pour obtenir le coefficient dominant

Macro-commandes de calculs artithmétiques

- `+`, `-` pour additionner et pour soustraire

`*`, `^` pour multiplier et pour élever à une puissance
`divide` pour vérifier la divisibilité d'un polynôme
`gcd` pour obtenir le plus grand commun diviseur de deux polynômes
`rem` pour obtenir le reste dans la division euclidienne de deux polynômes
`quo` pour obtenir le quotient dans la division euclidienne de deux polynômes

Macro-commandes de calculs mathématiques

`interp` pour obtenir le polynôme d'interpolation
`subs` pour évaluer un polynôme
`sum` pour obtenir la somme définie ou indéfinie

Macro-commandes de recherche de racines et de factorisation

`factor` pour obtenir la factorisation d'un polynôme sur un corps
`fsolve` pour obtenir des approximations des racines complexes (et réelles)
`realroot` pour obtenir la liste des intervalles renfermant les racines réelles
`roots` pour obtenir les racines exactes d'un polynôme sur un corps

Macro-commandes de simplification sur demande

`collect` pour regrouper les termes d'un polynôme
`expand` pour développer la distributivité de la multiplication sur l'addition
`sqrfree` square free factorization
`normal` pour réduire les expressions rationnelles
`sort` pour ordonner les termes d'un polynômes (plusieurs options offertes)

Exercices

No.1

Soit le polynôme $P = 2x^5 - 11x^4 + 14x^3 - 2x^2 + 12x - 9$. Est-ce que $\left(x + \frac{1}{2}\right)$ est un facteur du polynôme P ?

Solution proposée

Il suffit d'obtenir le reste avec le calcul de $P\left(-\frac{1}{2}\right)$ pour constater si ce reste est nul ou non. Si le reste est nul, nous pourrons affirmer que $\left(x + \frac{1}{2}\right)$ est effectivement un facteur du polynôme P.

```
> P:=2*x^5-11*x^4+14*x^3-2*x^2+12*x-9;
```

$$P := 2x^5 - 11x^4 + 14x^3 - 2x^2 + 12x - 9 \tag{10.1.1.1}$$

```
> 'P'(-1/2)=eval(P,x=-1/2);
```

$$P\left(-\frac{1}{2}\right) = -18 \tag{10.1.1.2}$$

Puisque $P\left(-\frac{1}{2}\right) \neq 0$, on conclut que $\left(x + \frac{1}{2}\right)$ n'est pas un facteur du polynôme P.

No.2

Soit P = $x^4 + a x^3 - a \mathrm{e} x + \pi$. Sachant que 1 est une racine du polynôme P, trouver le reste de la division de P par $x - 2$.

Solution proposée

Si 1 est une racine de P, on a nécessairement P(1) = 0 alors obtenons d'abord la valeur du paramètre « *a* » en résolvant l'équation $P(1) = 0$.

Posons l'équation P(1) = 0

```
> P:=x^4+a*x^3-exp(1)*a*x+Pi;
  Éq:=eval(P,x=1)=0;
```

$$P := x^4 + a x^3 - \mathrm{e} a x + \pi$$

$$\acute{E}q := 1 + a - \mathrm{e} a + \pi = 0 \quad (10.2.1.1)$$

Résolvons, pour *a*, l'équation Éq.

```
> solve(Éq,{a});
```

$$\left\{ a = \frac{1 + \pi}{\mathrm{e} - 1} \right\} \quad (10.2.1.2)$$

Après avoir assigné au paramètre « *a* » la valeur $\frac{1+\pi}{-1+\mathrm{e}}$, évaluons P(2). Ce qui donnera le reste de la division de P par $x - 2$.

```
> assign((10.2.1.2));
  'P'(a)=eval(P,x=2);
```

$$P\left(\frac{1+\pi}{\mathrm{e}-1}\right) = 16 + \frac{8\,(1+\pi)}{\mathrm{e}-1} - \frac{2\,\mathrm{e}\,(1+\pi)}{\mathrm{e}-1} + \pi \quad (10.2.1.3)$$

```
> ``=normal(rhs((10.2.1.3)));
```

$$= -\frac{\mathrm{e}\pi - 7\pi - 14\,\mathrm{e} + 8}{\mathrm{e}-1} \quad (10.2.1.4)$$

Dans la condition de l'énoncé, le reste cherché est $-\frac{\mathrm{e}\pi - 14\,\mathrm{e} - 7\pi + 8}{\mathrm{e}-1}$.

```
> a:='a':
```

No.3

Soit P1 = $3x^4 - x^3 + x^2 + 2$ et P2 = $x^2 + x + 2$. Effectuer la division P1 ÷ P2 et exprimer le résultat sous la forme de l'algorithme de la division. Utiliser seulement la marcro-commande `quo`.

Solution proposée

```
> P1:=3*x^4-x^3+x^2+2;
  P2:=x^2+x+2;
```

$$P1 := 3x^4 - x^3 + x^2 + 2$$

$$P2 := x^2 + x + 2 \quad (10.3.1.1)$$

```
> P1=P2*quo(P1,P2,x,'Reste')+Reste;
```

$$3x^4 - x^3 + x^2 + 2 = (x^2 + x + 2)\,(3x^2 - 4x - 1) + 9x + 4 \quad (10.3.1.2)$$

Réponse: $3x^4 - x^3 + x^2 + 2 = (x^2 + x + 2)\,(3x^2 - 4x - 1) + 9x + 4$

No.4

Même question que celle de l'exemple 3 mais en utilisant seulement la procédure division synthétique **div_syn**.

Solution proposée

Ce n'est pas possible car le diviseur est un polynôme du deuxième degré. On ne peut effectuer la division synthétique seulement lorsque le diviseur est du premier degré de la forme $(x-a)$.

No.5

Soit P1 = $3x^3 - x^2 + 6x - 5$ et $P2 = 3x - 1$. Effectuer la division P1÷ P2 et exprimer le résultat sous la forme de l'algorithme de la division. Utiliser seulement la procédure division synthétique **div_syn**.

Solution proposée

Recopions la procédure div_syn.

```
> div_syn:=proc(p::polynom,var::name,x0::complexcons)
   local A,a,k,n,Q,x;
   x:=var;
   if degree(p)=-infinity then
    printf(" ");
    printf("\n  Le reste de la division de %a par (%a) est %a",p,x-x0,p);
    printf("\n  Le polynôme quotient est 0")
   else
    A:=[seq(coeff(p,x,k),k=0..degree(p))];
    n:=nops(A);
    Q[1]:=A[n];
    for k from 1 to n-1 do
     Q[k+1]:=eval(Q[k]*x0+a[n-k-1],a[n-k-1]=A[n-k])
     od;
   printf(" ");
   printf("\n  Le reste de la division de (%a) par (%a) est %a",p,x-x0,
  sort(Q[k]));
   printf("\n  Le polynôme quotient est (%a)",sort(eval(sum(Q['k']*x^'n-
  k-1','k'=1..n-1))))
   fi
  end proc:
```

```
> P1:=3*x^3-x^2+6*x-5;
```

$$P1 := 3x^3 - x^2 + 6x - 5 \quad \textbf{(10.5.1.1)}$$

```
> div_syn(P1,x,1/3);

  Le reste de la division de (3*x^3-x^2+6*x-5) par (x-1/3) est -3
  Le polynôme quotient est (3*x^2+6)
```

Alors, on peut écrire

```
> P1=(x-1/3)*(3*x^2+6)-3;
```

$$3x^3 - x^2 + 6x - 5 = \left(x - \frac{1}{3}\right)\left(3x^2 + 6\right) - 3 \quad \textbf{(10.5.1.2)}$$

```
> ``=normal((x-1/3)*(3*x^2+6))+3;
```

$$= (3x - 1)\left(x^2 + 2\right) + 3 \quad \textbf{(10.5.1.3)}$$

No.6

Soit P = $x^5 - 3x^4 - x^3 + 11x^2 - 12x + 4$. Quel est l'ordre de multiplicité de 1 ? Répondre en utilisant seulement la procédure **div_syn** dans tous vos développements.

Solution proposée

Utilisons la division synthétique pour déterminer le reste de la division de P par $(x-1)$.

```
> P:=x^5-3*x^4-x^3+11*x^2-12*x+4;
```

$$P := x^5 - 3x^4 - x^3 + 11x^2 - 12x + 4 \qquad (10.6.1.1)$$

```
> div_syn(P,x,1);

  Le reste de la division de (x^5-3*x^4-x^3+11*x^2-12*x+4) par (x-1) est 0
  Le polynôme quotient est (x^4-2*x^3-3*x^2+8*x-4)
```

Puisque P(1) = 0, l'ordre de multiplicité de 1 est au moins 1. Divisons de quotient Q1 = $x^4 - 2x^3 - 3x^2 + 8x - 4$ par $(x-1)$.

```
> Q1:=x^4-2*x^3-3*x^2+8*x-4;
  div_syn(Q1,x,1);
```

$$Q1 := x^4 - 2x^3 - 3x^2 + 8x - 4$$

```
  Le reste de la division de (x^4-2*x^3-3*x^2+8*x-4) par (x-1) est 0
  Le polynôme quotient est (x^3-x^2-4*x+4)
```

Ce qui montre que l'ordre de mutiplicité de 1 est au moins 2. Divisons ce nouveau quotient Q2 = $x^3 - x^2 - 4x + 4$ par $(x-1)$.

```
> Q2:=x^3-x^2-4*x+4;
  div_syn(Q2,x,1);
```

$$Q2 := x^3 - x^2 - 4x + 4$$

```
  Le reste de la division de (x^3-x^2-4*x+4) par (x-1) est 0
  Le polynôme quotient est (x^2-4)
```

Ce qui montre que l'ordre de mutiplicité de 1 est au moins 3.
Il n'est pas utile de faire intervenir Maple pour le troisième quotient Q3 = $x^2 - 4$. Il est clair que Q3(1) = $-3 \neq 0$, et donc que Q3 n'est pas divisible par $(x-1)$.

Puisque le reste est non nul à la quatrième division, l'ordre de multiplicité de la racine 1 est égale à 3.

Ce qui nous permet d'écrire l'égalité

$$x^5 - 3x^4 - x^3 + 11x^2 - 12x + 4 = (x-1)^3(x^2-4)$$

No.7

Montrer que le polynôme P = $x^n + a^n$ où $a \neq 0$ et $n \in \mathbb{N}$ est divisible par $(x+a)$ si n est impair.

Solution proposée

Il sufffit de montrer que $P(-a) = 0$, pour tout $n \in \mathbb{N}$ impair. En effet, si on montre, lorsque n est impair, que $P(-a) = 0$, on montre alors que $(x-a)$ est un facteur de P et donc que $(x-a)$ divise le polynôme P.

```
> P:=x^n+a^n;
```

$$P := x^n + a^n \qquad (10.7.1.1)$$

Obtenons le reste de la division de $x^n + a^n$ par $x + a$.

```
> Eval(P,x=-a)=eval(P,x=-a);
```

(10.7.1.2)

$$(x^n + a^n)\Big|_{x=-a} = (-a)^n + a^n \quad (10.7.1.2)$$

Simplifions ensuite le reste sur la base que n est impair.

```
> ``=simplify(rhs((10.7.1.2))) assuming n::odd;
```

$$= 0 \quad (10.7.1.3)$$

Ce qui montre que le reste est nul lorsque n est impair. Donc, $x + a$ divise $x^n + a^n$ lorsque n est impair.

No.8

Donner la forme développée d'un polynôme P du troisième degré dont les racines sont $\sqrt{2}$, $-\sqrt{2}$ et 3 et pour lequel le coefficient dominant est égal à 2.

Solution proposée

Il suffit de poser directement $P = 2\left(x - \sqrt{2}\right)\left(x + \sqrt{2}\right)(x - 3)$.

```
> P:=2*(x-sqrt(2))*(x+sqrt(2))*(x-3);
```

$$P := 2\left(x - \sqrt{2}\right)\left(x + \sqrt{2}\right)(x - 3) \quad (10.8.1.1)$$

```
> 'P(x)'=expand(P);
```

$$P(x) = 2x^3 - 6x^2 - 4x + 12 \quad (10.8.1.2)$$

Le polynôme à déterminer est donc $2x^3 - 6x^2 - 4x + 12$.

No.9

Trouver une fonction polynomiale du deuxième degré dont le graphique passe par les points (1,7), (-1,-1) et (-2,2).

Solution proposée

Soit f la fonction définie par $f(x) = ax^2 + bx + c$. Déterminons les valeurs de a, b et c sachant que $f(1) = 7, f(-1) = -1$ et que $f(-2) = 2$.

```
> f:= x->a*x^2+b*x+c;
```

$$f := x \mapsto a \cdot x^2 + b \cdot x + c \quad (10.9.1.1)$$

Posons le système suivant.

```
> Éq1:=f(1)=7:
  Éq2:=f(-1)=-1:
  Éq3:=f(-2)=2:
  Système:={Éq1,Éq2,Éq3};
```

$$Système := \{a - b + c = -1, a + b + c = 7, 4a - 2b + c = 2\} \quad (10.9.1.2)$$

Résolvons ensuite ce système.

```
> Solution:=solve(Système,{a,b,c});
  assign(Solution);
  'f'(x)=f(x);
```

$$Solution := \left\{a = \frac{7}{3}, b = 4, c = \frac{2}{3}\right\}$$

$$f(x) = \frac{7}{3}x^2 + 4x + \frac{2}{3} \quad (10.9.1.3)$$

Réponse: La fonction f cherchée est définie par $f(x) = \frac{7x^2}{3} + 4x + \frac{2}{3}$.

Pour contrôler notre réponse, illustrons, dans un même graphique, les trois points donnés ainsi que la courbe d'équation $y = \frac{7x^2}{3} + 4x + \frac{2}{3}$.

```
> Opts:=symbol=solidcircle,symbolsize=15,color="Niagara 2":
  Point_1:=pointplot([1,7],Opts):
  Point_2:=pointplot([-1,-1],Opts):
  Point_3:=pointplot([-2,2],Opts):
  Courbe:=plot([x,7/3*x^2+4*x+2/3,x=-3..2],color="Niagara 1",
  thickness=1):
  display([Courbe,Point_1,Point_2,Point_3],scaling=constrained,
  view=[-3..2,-2..8]);
```




```
> unassign('a','b','c');
```

No.10

Modifier légèrement la procédure div_syn afin qu'elle donne comme résultat la division euclidienne d'un polynôme P divisé par $(x - a)$. Nommer votre nouvelle procédure div_euclid.

Suggestion: Utiliser les crochets au lieu des parenthèses rondes afin de contourner le mécanisme de la simplification automatique de Maple.

Voici, par exemple, le résultat de la procédure div_euclid pour $(6x^5 - 20x^3 - 7x + 10) \div (x + 2)$:

$$6x^5 - 20x^3 - 7x + 10 = [x + 2]\,[6x^4 - 12x^3 + 4x^2 - 8x + 9] + [-8]$$

Solution proposée

```
> div_euclid:=proc(p::polynom,var::name,x0::complexcons)
   local A,a,k,n,Q,Quotient,Reste,x;
   x:=var;
   if degree(p)=-infinity then
    return(p=[x-x0]*[0]+[0]);
   else
    A:=[seq(coeff(p,x,k),k=0..degree(p))];
    n:=nops(A);
    Q[1]:=A[n];
```

```
   for k from 1 to n-1 do
    Q[k+1]:=eval(Q[k]*x0+a[n-k-1],a[n-k-1]=A[n-k])
    od;
  fi;
  Quotient:=radnormal(sum(Q['k']*x^'n-k-1','k'=1..n-1));
  Reste:=simplify(Q[k]);
  return(p=[x-x0]*[Quotient]+[Reste])
  end proc:
```

```
> P:=6*x^5-20*x^3-7*x+10;
```

$$P := 6x^5 - 20x^3 - 7x + 10 \quad \textbf{(10.10.1.1)}$$

```
> div_euclid(P,x,-2);
```

$$6x^5 - 20x^3 - 7x + 10 = [x+2]\,[6x^4 - 12x^3 + 4x^2 - 8x + 9] + [-8] \quad \textbf{(10.10.1.2)}$$